
Abstract. Equational unification and matching are fundamental mech-
anisms in many automated deduction applications. Supporting them effi-
ciently for as wide as possible a class of equational theories, and in a
typed manner supporting type hierarchies, benefits many applications;
but this is both challenging and nontrivial. We present Maude 3.2’s effi-
cient support of these features as well as of symbolic reachability analysis
of infinite-state concurrent systems based on them.

1 Introduction

Unification is a key mechanism in resolution [41] and paramodulation-based
[36] theorem proving. Since Plotkin’s work [40] on equational unification, i.e.,

E-unification modulo an equational theory E, it is widely used for increased
effectiveness. Since Walther’s work [47] it has been well understood that typed
E-unification, exploiting types and subtype hierarchies, can drastically reduce a
prover’s search space. Many other automated deduction applications use typed
E-unification as a key mechanism, including, inter alia: (i) constraint logic pro-
gramming, e.g., [12,23]; (ii) narrowing-based infinite-state reachability analysis
and model checking, e.g., [6,35]; (iii) cryptographic protocol analysis modulo
algebraic properties, e.g., [8,19,28]; (iv) partial evaluation, e.g., [4,5]; and (v)
SMT solving, e.g., [32,48]. The special case of typed E-matching is also a key
component in all the above areas as well as in: (vi) E-generalization (also called
anti-unification), e.g., [1,2]; and (vii) E-homeomorphic embedding, e.g., [3].

Maximizing the scope and effectiveness of typed E-unification and E-
matching means efficiently supporting as wide a class of theories E as possible.
Such efficiency crucially depends on both efficient algorithms (and their com-
binations) and —since the number of E-unifiers may be large— on computing
complete minimal sets of solutions to reduce the search space. The recent Maude
3.2 release1 provides this kind of efficient support for typed E-unification and
E-matching in three, increasingly more general classes of theories E:

1. Typed B-unification and B-matching, where B is any combination of asso-
ciativity (A) and/or commutativity (C) and/or unit element (U) axioms.

2. Typed E ∪ B-unification and matching in the user-definable infinite class of
theories E ∪B with B as in (1), and E ∪B having the finite variant property
(FVP) [13,21].

3. Typed E ∪B-unification for the infinite class of user-definable theories E ∪B
with B as in (1), and E confluent, terminating, and coherent modulo B.

For classes (1) and (2) the set of B- (resp. E ∪ B-) unifiers is always complete,
minimal and finite, except for the AwoC case when B contains an A but not C
axiom for some binary symbol f .2 The typing is order-sorted [22,29] and thus
contains many-sorted and unsorted B- (resp. E∪B-) unification as special cases.
For class (3), Maude enumerates a possibly infinite complete set of E∪B-unifiers,
with the same AwoC exception on B. We discuss new features for classes (1)–(2),
and a new narrowing modulo E ∪B-based symbolic reachability analysis feature
for infinite-state systems specified in Maude as rewrite theories (Σ,E ∪ B,R)
with equations E ∪ B in class (2) and concurrent transition rules R. In Sect. 5
we discuss various applications that can benefit from these new features.

In comparison with previous Maude tool papers reporting on new features
—the last one was [16]— the new features reported here include: (i) computing
minimal complete sets of most general B- (resp. E ∪ B-) unifiers for classes (1)
and (2) except for the AwoC case; (ii) a new E ∪ B-matching algorithm for
class (2); and (iii) a new symbolic reachability analysis for concurrent systems

1 Publicly available at http://maude.cs.illinois.edu.
2 In the AwoC case, Maude’s algorithms are optimized to favor many commonly occur-

ring cases where typed A-unification is finitary, and provides a finite set of solutions
and an incompleteness warning outside such cases (see [18]).

http://maude.cs.illinois.edu

based on narrowing with transition rules modulo equations E ∪ B in class (2)
enjoying powerful state-space reduction capabilities based on the minimality and
completeness feature (i) and on “folding” less general symbolic states into more
general ones through subsumption. Section 3.1 shows the importance of the new
E ∪ B-matching algorithm for efficient computation of minimal E ∪ B-unifiers.

Notation, Strict-B-Coherence, and FVP. For notation involving either term
positions, p ∈ pos(t), t|p, t[t′]p, or substitutions, tθ, θμ, see [14]. Equations
(u = v) ∈ E oriented as rules (u → v) ∈ −→

E are strictly coherent modulo axioms
B iff (t =B t′ ∧ t →−→

E ,B
w) ⇒ ∃w′(t →−→

E ,B
w′ ∧ w =B w′), where t →−→

E ,B
w

iff ∃(u → v) ∈ −→
E , ∃θ, ∃p ∈ pos(t)(uθ =B t|p ∧ w = t[vθ]p). For (Σ,E ∪ B) an

equational theory with
−→
E confluent, terminating and strictly coherent modulo

B, (1) an
−→
E ,B-t-variant is a pair (v, θ) s.t. v = (tθ)!−→

E ,B
∧ θ = θ!−→

E ,B
, where

u!−→
E ,B

(resp. θ!−→
E ,B

) denotes the
−→
E ,B-normal form of u, resp. θ; (2) for

−→
E ,B-t-

variants (v, θ), (u, μ), the more general relation (v, θ) �B (u, μ) holds iff ∃γ(u =B

vγ ∧ θγ =B μ); (3) (Σ,E ∪ B) is FVP [13,21] iff any Σ-term t has a finite set
of most general

−→
E ,B-t-variants. Footnote 5 explains how FVP can be checked.

2 Complete and Minimal Order-Sorted B-Unifiers

Throughout the paper we use the following equational theory E ∪ B of the
Booleans as a running example (with self-explanatory, user-definable syntax3):

fmod BOOL-FVP is protecting TRUTH-VALUE .
op _and_ : Bool Bool -> Bool [assoc comm] .
op _xor_ : Bool Bool -> Bool [assoc comm] .
op not_ : Bool -> Bool .
op _or_ : Bool Bool -> Bool .
op _<=>_ : Bool Bool -> Bool .
vars X Y Z W : Bool .

eq X and true = X [variant] .
eq X and false = false [variant] .
eq X and X = X [variant] .
eq X and X and Y = X and Y [variant] . *** AC extension
eq X xor false = X [variant] .
eq X xor X = false [variant] .
eq X xor X xor Y = Y [variant] . *** AC extension
eq not X = X xor true [variant] .
eq X or Y = (X and Y) xor X xor Y [variant] .
eq X <=> Y = true xor X xor Y [variant] .

endfm

3 This module imports Maude’s TRUTH-VALUE module and the command “set include

BOOL off .” must be typed before the module to avoid default importation of BOOL.

The axioms B are the associativity-commutativity (AC) axioms for xor and and
(specified with the assoc comm attributes). The equations E are terminating and
confluent modulo B [42]. To achieve strict B-coherence [30], the needed AC-
extensions [39] are added —for example, the AC-extension of X xor X = false
is X xor X xor Y = Y. The equations E for xor and and define the theory of
Boolean rings, except for the missing4 distributivity equation X and (Y xor Z)
= (X and Y) xor (X and Z). The remaining equations in E define or, not and
<=> as definitional extensions. The variant attribute declares that the equation
will be used for folding variant narrowing [21]. The theory is FVP,5 in class (2).
In this section we will consider B-unification (for B = AC) using this example.
E ∪ B-unification for the same example will be discussed in Sect. 3.

For B any combination of associativity and/or commutativity and/or iden-
tity axioms, Maude’s unify command computes a complete finite set of most
general B-unifiers, except for the AwoC case. The new irredundant unify com-
mand always returns6 a finite, complete and minimal set of B-unifiers, except
for the AwoC case. The output of unify for the equation below can be found
in [10, §13].

Maude> irredundant unify X and not Y and not Z =? W and Y and not X .

Decision time: 0ms cpu (0ms real)

Unifier 1 Unifier 2

X --> #1:Bool and #2:Bool X --> #2:Bool

Z --> #1:Bool and #2:Bool Z --> #1:Bool

Y --> #1:Bool Y --> #2:Bool

W --> #2:Bool and not #1:Bool W --> not #1:Bool

3 E ∪ B-Unification and Matching for FVP Theories

It is a general result from [21] that if E ∪B is FVP and B-unification is finitary,
then E ∪ B-unification is finitary and a complete finite set of E ∪ B-unifiers
can be computed by folding variant narrowing [21]. Furthermore, assuming that
TΣ/E,s is non-empty for each sort s, a finitary E ∪ B-unification algorithm
automatically provides a decision procedure for satisfiability of any positive (the
∧,∨-fragment) quantifier-free formula ϕ in the initial algebra TΣ/E , since ϕ can
be put in DNF, and a conjunction of equalities Γ is satisfiable in TΣ/E iff Γ is
E ∪ B-unifiable.

Since for our running example BOOL-FVP the equations E∪B are FVP and B-
unification (in this case B = AC) is finitary, all this has useful consequences for
4 By missing distributivity, this theory is weaker than the theory of Boolean rings.

Nevertheless, its initial algebra TΣ/E∪B is exactly the Booleans on {true,false}
with the standard truth tables for all connectives. Thus, all equations provable in
Boolean algebra hold in TΣ/E∪B , including the missing distributivity equation.

5 This can be easily checked in Maude by checking the finiteness of the variants for
each f(X), resp. f(X,Y ), for each unary, resp. binary, symbol f in BOOL-FVP using
the get variants command; see [9] for a theoretical justification of this check.

6 Fresh variables follow the form #1:Bool.



Equational Unification and Matching, and Symbolic Reachability Analysis 533

BOOL-FVP. Indeed, TΣ/E∪B is exactly the Booleans7 on {true,false} with the
well-known truth tables for and, xor, not, or and <=>. This means that E ∪ B-
unification provides a Boolean satisfiability decision procedure for a Boolean
expression u on such symbols, namely, u is Boolean satisfiable iff the equation
u = true is E∪B-unifiable. Furthermore, a ground assignment ρ to the variables
of u is a satisfying assignment for u iff there exists an E∪B-unifier α of u = true
and a ground substitution δ such that ρ = αδ. For the same reasons, u is a
Boolean tautology iff the equation u = false has no E ∪ B-unifiers.

A complete, finite set of E ∪ B-unifiers can be computed with Maude’s
variant unify command whenever E ∪ B is FVP, except for the AwoC case.
Instead, the new8 filtered variant unify command computes a finite, com-
plete and minimal set of E ∪ B-unifiers, which can be considerably smaller
than that computed by variant unify. For our BOOL-FVP example, filtered
variant unify gives us a Boolean satisfiability decision procedure plus a sym-
bolic specification of satisfying assignments. Such a procedure is not practical: it
cannot compete with standard SAT-solvers; but that was never our purpose: our
purpose here is to illustrate with simple examples how E ∪ B-unification works
for the infinite class of user-definable FVP theories E ∪ B, of which BOOL-FVP
is just a simple example; dozens of other examples can be found in [32].

The difference between the variant unify and the new filtered variant
unify command is illustrated with the following example; its unfiltered output
can be found in [10, §14]. Note that the single E ∪ B-unifier gives us a compact
symbolic description of this Boolean expression’s satisfying assignments.

Maude> filtered variant unify (X or Y) <=> Z =? true .
rewrites: 3224 in 12765ms cpu (14776ms real) (252 rewrites/second)

Unifier 1
X --> #1:Bool xor #2:Bool
Y --> #1:Bool
Z --> #2:Bool xor (#1:Bool and (#1:Bool xor #2:Bool))

No more unifiers.
Advisory: Filtering was complete.

The computation of a minimal set of E ∪ B-unifiers relies on filtering by E ∪ B-
matching between two E ∪ B-unifiers, as explained in the following section.

3.1 FVP E ∪ B-Matching and Minimality of E ∪ B-Unifiers

By definition, a term u E ∪ B-matches another term v iff there is a substitution
γ such that u =E∪B vγ. Besides the existing match command modulo axioms
7 Each connective’s truth table can be checked with Maude’s reduce command. Actu-

ally, need only check and and xor (other connectives are definitional extensions).
8 In Maude, different command names are used to emphasize different algorithms.

The word ‘filtered’ is used instead of ‘irredundant’ because irredundancy is not
guaranteed in the AwoC case.

B, Maude’s new variant match command computes a complete, minimal set
of E ∪ B-matching substitutions for any FVP theory E ∪ B in class (2), except
for the AwoC case. Such an algorithm could always be derived from an E ∪ B-
unification algorithm by replacing u by u, where all variables in u are replaced
by fresh constants in u, and computing the E ∪ B-unifiers of u = v. But a more
efficient special-purpose algorithm has been designed and implemented for this
purpose. E ∪ B-matching algorithms are automatically provided by Maude for
any user-definable theory in class (2) with the variant match command.

Maude> variant match in BOOL-FVP : Z and W <=? X .
rewrites: 12 in 21ms cpu (27ms real) (545 rewrites/second)

Matcher 1 Matcher 2 Matcher 3
Z --> true Z --> X Z --> X
W --> X W --> true W --> X

This is a good moment to ask and answer a relevant question: Why is com-
puting a complete minimal set of E ∪ B-unifiers for a unification problem Γ ,
where E ∪ B is an FVP theory in class (2) except for the AwoC case, non-
trivial? We first need to explain how minimality is achieved. Suppose that α
and β are two E ∪ B-unifiers of a system of equations Γ with, say, typed vari-
ables x1, . . . , xn. We then say that α is more general than β modulo E ∪ B,
denoted α �E∪B β, iff there is a substitution γ such that for each xi, 1 ≤ i ≤ n,
γ(α(xi)) =E∪B β(xi). But this exactly means that the vector [β(x1), . . . , β(xn)]
E ∪ B-matches the vector [α(x1), . . . , α(xn)] with E ∪ B-matching substitution
γ. A complete set of E ∪ B-unifiers of Γ is by definition minimal iff for any two
different unifiers α and β in it we have α ��E∪B β and β ��E∪B α, i.e., the two
associated E ∪ B-matching problems fail.

What is nontrivial is computing a minimal complete set of E ∪ B-unifiers
efficiently. One could do so inefficiently by simulating E ∪ B-matching with E ∪
B-unification, and more efficiently by using an E∪B-matching algorithm. Maude
achieves still greater efficiency by directly computing the α �E ∪ B β relation.
The key difference between the variant unify command and the new filtered
variant unify command is that the second computes a E ∪ B-minimal set of
E∪B-unifiers of Γ using the α �E∪B β relation, whereas the first only computes
a set of B-minimal E ∪B-unifiers of Γ using the cheaper α �B β relation. There
are three ideas we use to make it fast in practice: (i) variant matching is faster
than variant unification because one side is variable-free; (ii) enumerating the
variant matchers between two variant unifiers is far more expensive than checking
existence of a matcher; and (iii) variant unifiers are discarded on-the-fly avoiding
further narrowing steps and computation.

4 Narrowing-Based Symbolic Reachability Analysis

In Maude, concurrent systems are specified in so-called system modules as rewrite
theories of the form: R = (Σ,G,R), where G is an equational theory either of the

form B in class (1), or E∪B in classes (2) or (3), and R are the system transition
rules, specified as rewrite rules. When the theory R is topmost, meaning that the
rules R rewrite the entire state, narrowing with rules R modulo the equations
G is a complete symbolic reachability analysis method for infinite-state systems
[35]. That is, given a term u with variables −→x , representing a typically infinite
set of initial states, and another term v with variables −→y , representing a possibly
infinite set of target states, narrowing can answer the question: can an instance
of u reach an instance of v? That is, does the formula ∃−→x ,−→y u →∗ v hold in
R? Note that, if the complement of a system invariant I can be symbolically
described as the set of ground instances of terms in a set {v1, . . . , vn} of pattern
terms, then narrowing provides a semi-decision procedure for verifying whether
the system specified by R fails to satisfy I starting from an initial set of states
specified by u. Namely, I holds iff no instance of any vi can be reached from
some instance of u.

Assuming G is in class (1) or (2), Maude’s vu-narrow command implements
narrowing with R modulo G by performing G-unification at each narrowing
step. However, the number of symbolic states that need to be explored can be
infinite. This means that if no solution exists for the narrowing search, Maude
will search forever, so that only depth-bounded searches will terminate. The great
advantage of the new {fold} vu-narrow {filter,delay} command is that it
performs a powerful symbolic state space reduction by: (i) removing a newly
explored symbolic state v′ if it E ∪B-matches a previously explored state v and
replacing transition with target v′ by transitions with target v; and (ii) using
minimal sets of E ∪B-unifiers for each narrowing step and for checking common
instances between a newly explored state and the target term (ensured by words
filter and delay). This can make the entire search space finite and allow full
verification of invariants for some infinite-state systems. Consider the following
Maude specification of Lamport’s bakery protocol.

mod BAKERY is

sorts Nat LNat Nat? State WProcs Procs .

subsorts Nat LNat < Nat? . subsort WProcs < Procs .

op 0 : -> Nat .

op s : Nat -> Nat .

op [_] : Nat -> LNat . *** number-locking operator

op < wait,_> : Nat -> WProcs .

op < crit,_> : Nat -> Procs .

op mt : -> WProcs . *** empty multiset

op __ : Procs Procs -> Procs [assoc comm id: mt] . *** union

op __ : WProcs WProcs -> WProcs [assoc comm id: mt] . *** union

op _|_|_ : Nat Nat? Procs -> State .

vars n m i j k : Nat . var x? : Nat? . var PS : Procs . var WPS : WProcs .

rl [new]: m | n | PS => s(m) | n | < wait,m > PS [narrowing] .

rl [enter]: m | n | < wait,n > PS => m | [n] | < crit,n > PS [narrowing] .

rl [leave]: m | [n] | < crit,n > PS => m | s(n) | PS [narrowing] .

endm

The states of BAKERY have the form “m | x? | PS” with m the ticket-dispensing
counter, x? the (possibly locked) counter to access the critical section, and PS a
multiset of processes either waiting or in the critical section. BAKERY is infinite-
state: [new] creates new processes, and the counters can grow unboundedly.
When a waiting process enters the critical section with [enter], the second
counter n is locked as [n]; and it is unlocked and incremented when it leaves
it with [leave]. The key invariant is mutual exclusion. Note that the term
“i | x? | < crit, j > < crit, k > PS” describes all states in the comple-
ment of mutual exclusion states. Without the fold option, narrowing does not
terminate, but with the following command we can verify that BAKERY satisfies
mutual exclusion, not just for the initial state “0 | 0 | mt”, but for the much
more general infinite set of initial states with waiting processes only “m | n |
WPS”.

Maude> {fold} vu-narrow {filter,delay}
m | n | WPS =>* i | x? | < crit, j > < crit, k > PS .

No solution.
rewrites: 4 in 1ms cpu (1ms real) (2677 rewrites/second)

The new vu-narrow {filter,delay} command can achieve dramatic state
space reductions over the previous vu-narrow command by filtering E ∪ B-
unifiers. This is illustrated by a simple cryptographic protocol example in [10,
§15] exploiting the unitary nature of unification in the exclusive-or theory [24].

5 Applications and Conclusion

Maude can be used as a meta-tool to develop new formal tools because: (i) its
underlying equational and rewriting logics are logical —and reflective meta-
logical— frameworks [7,27,46]; (ii) Maude’s efficient support of logical reflection
through its META-LEVEL module; (iii) Maude’s rewriting, search, model checking,
and strategy language features [11,15]; and (iv) Maude’s symbolic reasoning
features [15,33], the latest reported here. We refer to [11,15,31,33] for references
on various Maude-based tools. Many of them can benefit from these new features.

By way of example we mention some areas ready to reap such benefits: (1)
Formal Analysis of Cryptographic Protocols. The new features can yield substan-
tial improvements to tools such as Maude-NPA [19], Tamarin [28] and AKISS [8].
(2) Model Checking of Infinite-State Systems. The narrowing-based LTL sym-
bolic model checker reported in [6,20], and the addition of new symbolic capa-
bilities to Real-Time Maude [37,38] can both benefit from the new features. (3)
SMT Solving. In Sect. 3 we noted that FVP E∪B-unification makes satisfiability
of positive QF formulas in TΣ/E∪B decidable. Under mild conditions, this has
been extended in [32,44] to a procedure for satisfiability in TΣ/E∪B of all QF
formulas which will also benefit from the new features. (4) Theorem Proving.
The new Maude Inductive Theorem Prover under construction [34], as well as

Maude’s Invariant Analyzer [43] and Reachability Logic Theorem Prover [45] all
use equational unification and narrowing modulo equations; so all will benefit
from the new features. (5) Theory Transformations based on equational unifi-
cation, e.g., partial evaluation [4], ground confluence methods [17] or program
termination methods [25,26] could likewise become more efficient.

In conclusion, we have presented and illustrated with examples new equa-
tional unification and matching, and symbolic reachability analysis features in
Maude 3.2. Thanks to the above-mentioned properties (i)–(iv) of Maude as a
meta-tool, we hope that this work will encourage other researchers to use Maude
and its symbolic features to develop new tools in many different logics.

